

# Introduction to Neural Networks

## U. Minn. Psy 5038

### Linear discriminant

#### Initialize

Read in Statistical Add-in packages:

```
Off[General::"spell1"];  
<< "MultivariateStatistics`";  
<< "ComputationalGeometry`"
```

---

## Discriminant functions

Let's build our geometric intuitions of what a simple perceptron unit does by viewing it from a more formal point of view. Perceptron learning is an example of nonparametric statistical learning, because it doesn't require knowledge of the underlying probability distributions generating the data (such distributions are characterized by a relatively small number of "parameters", such as the mean and variance of a Gaussian distribution). Of course, how well it does will depend on the generative structure of the data. Much of the material below is covered in Duda and Hart (1978).

#### Linear discriminant functions: Two category case

A discriminant function,  $g(\mathbf{x})$  divides input space into two category regions depending on whether  $g(\mathbf{x}) > 0$  or  $g(\mathbf{x}) < 0$ . (We've switched notation,  $\mathbf{x} = \mathbf{f}$ ). The linear case corresponds to the simple perceptron unit we studied earlier:

$$g(\mathbf{x}) = \mathbf{w} \cdot \mathbf{x} + w_0$$

where  $\mathbf{w}$  is the weight vector and  $w_0$  is the (scalar) threshold (sometimes called bias, although this "bias" has nothing to do with statistical "bias").

Discriminant functions can be generalized, for example to quadratic decision surfaces:

$$g(\mathbf{x}) = w_0 + \sum_{i=1} w_i x_i + \sum_{i=1} \sum_{j=1} w_{ij} x_i x_j$$

where  $\mathbf{x} = \{x_1, x_2, x_3, \dots\}$ . We've seen how  $g(\mathbf{x}) = 0$  defines a decision surface which in the linear case is a hyperplane.

Suppose  $\mathbf{x}_1$  and  $\mathbf{x}_2$  are vectors, with endpoints sitting on the hyperplane, then their difference is a vector lying in the hyperplane

$$\begin{aligned} \mathbf{w} \cdot \mathbf{x}_1 + w_0 &= \mathbf{w} \cdot \mathbf{x}_2 + w_0 \\ \mathbf{w} \cdot (\mathbf{x}_1 - \mathbf{x}_2) &= 0 \end{aligned}$$

so the weight vector  $\mathbf{w}$  is normal to any vector lying in the hyperplane. Thus  $\mathbf{w}$  determines how the plane is oriented. The normal vector  $\mathbf{w}$  points into the region for which  $g(\mathbf{x}) > 0$ , and  $-\mathbf{w}$  points into the region for which  $g(\mathbf{x}) < 0$ .

Let  $\mathbf{x}$  be a point on the hyperplane. If we project  $\mathbf{x}$  onto the normalized weight vector  $\mathbf{x} \cdot \mathbf{w} / |\mathbf{w}|$ , we have the normal distance of the hyperplane from the origin equal to:

$$\mathbf{w} \cdot \mathbf{x} / |\mathbf{w}| = -w_0 / |\mathbf{w}|$$


Thus, the threshold determines the position of the hyperplane.

One can also show that the normal distance of  $\mathbf{x}$  to the hyperplane is given by:

$$g(\mathbf{x}) / |\mathbf{w}|$$

So we've seen that: 1) discriminant function divides the input space by a hyperplane decision surface; 2) The orientation of the surface is determined by the weight vector  $w$ ; 3) the location is determined by the threshold  $w_0$ ; 4) the discriminant function gives a measure of how far an input vector is from the hyperplane.


The figure summarizes the basic properties of the linear discriminant.



```

Clear[w1, w2, w, w0];
Manipulate[
  (* x0={2,1};*)
  w = {w1, w2}; wn = w / Norm[w];
  g[x_] := {w1, w2}.x + w0;
  gg = Plot[Tooltip[
    x2 /. Solve[{w1, w2}.{x1, x2} + w0 == 0, x2], "discriminant"], {x1, -1, 3}];
  ggg = Graphics[g[Dynamic[MousePosition["Graphics"]]]];
  Show[{gg, Graphics[Inset["g[x]=", {1.6, 2}]],
    Graphics[Inset[ToString[g[x0]], {2, 2}]], Graphics[
      {Tooltip[Arrow[{{0, 0}, w}], "w"], Tooltip[Arrow[{{0, 0}, (-w0 / Norm[w]) * wn}],
        "-w0/|w|"], Tooltip[Arrow[{{0, 0}, x0}], "x"],
      Tooltip[Arrow[{{x0, x0 - wn * g[x0] / Norm[w]}], "g(x)/|w|"}]},
    PlotRange -> {{-1, 3}, {-1, 3}}, AxesOrigin -> {0, 0}, Axes -> True,
    AspectRatio -> 1, ImageSize -> Small],
  {{w0, -2.5}, -6, 3}, {{w1, 1}, 0, 3}, {{w2, 2}, 0, 3},
  {{x0, {2, 1}}, Locator}]

```




## Multiple classes

Suppose there are  $c$  classes. There are a number of ways to define multiple class discriminant rules. One way that avoids undefined regions is:

$$g_i(\mathbf{x}) = \mathbf{w}_i \cdot \mathbf{x} + w_{i0}, \quad i = 1, \dots, c$$

Assign  $\mathbf{x}$  to the  $i$ th class if:  $g_i(\mathbf{x}) > g_j(\mathbf{x})$  for all  $j \neq i$ .



(Adapted from Duda & Hart, 1973)

It can be shown that this classifier partitions the input space into simply connected convex regions. This means that if you connect any two feature vectors belonging to the same class by a line, all points on the line are in the same class. Thus this linear classifier won't be able to handle problems for which there are disconnected clusters of features that all belong to the same class. Also, from a probabilistic perspective, if the underlying generative probability model for a given class has multiple peaks, this linear classifier won't do a good job either.

---

## Task-dependent Dimensionality reduction

### Motivation

Later, when we consider the problem of dimensionality reduction, we will take another look at hyperplanes. But here the idea will be to find hyperplanes onto which we can project our input data, and from there divide up the hyperplane into decision regions. The idea is that the original input space may be impractically huge, but if we can find a subspace (hyperplane) that preserves the distinctions between categories as well as possible, we can make our decisions in smaller space. We will derive the Fisher linear "discriminant".

This is closely related to the psychology idea of finding "distinctive" features. E.g. consider bird identification. If I want to discriminate cardinals from other birds in my backyard, I can make use of the fact that (males) cardinals may be the only birds that are red. So even tho' the image of a bird can have lots of dimensions, if I project the image on to the "red" axis, I can do fairly well with just one number. How about male vs. female human faces?

## Fisher's linear "discriminant"

### Generative model: two nearby gaussian classes

Define two bivariate base distributions

```
(ar = {{1, 0.99}, {0.99, 1}};
ndista = MultinormalDistribution[{0, -1}, ar];)
(br = {{1, .9}, {.9, 2}};
ndistb = MultinormalDistribution[{0, 1}, br];)
```

Find the expression for the probability distribution function of ndista

```
pdf = PDF[ndista, {x1, x2}]
```

$$1.12822 e^{\frac{1}{2}(-x_1(50.2513 x_1 - 49.7487(x_2 + 1)) - (x_2 + 1)(50.2513(x_2 + 1) - 49.7487 x_1))}$$

Use Mean[] and CovarianceMatrix[ndista] to verify the population mean and the covariance matrix of ndista

```
Mean[ndista]
```

```
{0, 1}
```


```
Covariance[ndista]
```

```
{{1, 0.99}, {0.99, 1}}
```

Try different covariant matrices. Should they be symmetric? Constraints on the determinant of ar, br?  
Make a contour plot of the PDF ndista

```
pdfa = PDF[ndista, {x1, x2}];
```


```
ContourPlot[pdfa, {x1, -3, 3}, {x2, -3, 3}, PlotPoints -> 64, PlotRange -> All]
```



```

nsamples = 500;
a = Table[Random[ndista], {nsamples}];
ga = ListPlot[a, PlotRange → {{-8, 8}, {-8, 8}},
  AspectRatio → 1, PlotStyle → Hue[0.2`], DisplayFunction → Identity];
b = Table[Random[ndistb], {nsamples}];
gb = ListPlot[b, PlotRange → {{-8, 8}, {-8, 8}},
  AspectRatio → 1, PlotStyle → Hue[0.6`], DisplayFunction → Identity];
Show[ga, gb, DisplayFunction → $DisplayFunction]

```



Use `Mean[]` to find the *sample* mean of `b`. What is the *sample covariance* of `b`?

```
Mean[b]
```

```
{-0.0634543, 0.930223}
```

```
Covariance[b]
```

```
{{1.05063, 0.957104}, {0.957104, 1.98301}}
```

Try out different projections of the data by varying the slope ( $m$ ) of the discriminant line

```
m = -2 / 3;
```

```
wnvec = {1, m} / Sqrt[1 + m^2];
```

```
{{x, y}} . {n1, n2}
```

```
Map[#1 * {n1, n2} &, {{x, y}} . {n1, n2}]
```


```
{n1 x + n2 y}
```

```
{n1 (n1 x + n2 y), n2 (n1 x + n2 y)}
```

```

aproj = (#1 wnvec &) /@ (a.wnvec);
gaproj = ListPlot[aproj, AspectRatio -> 1,
  PlotStyle -> Hue[0.3`], DisplayFunction -> Identity];
bproj = (#1 wnvec &) /@ (b.wnvec);
gbproj = ListPlot[bproj, AspectRatio -> 1,
  PlotStyle -> Hue[0.7`], DisplayFunction -> Identity];
Show[ga, gb, gaproj, gbproj, DisplayFunction -> $DisplayFunction,
  AspectRatio -> Automatic]

```



By trial and error, find a value of  $m$  that separates the classes well along the projection line. Plot out the marginal distributions relative to this line.

Calculate the "signal-to-noise" ratio along the projection line. Do this by taking the difference between the means divided by the square root of the product of the standard deviations along the line.

```
Mean[aproj]
```

```
{0.473876, -0.315917}
```

## Theory for simple 2-class case

(see Duda and Hart for general case)

A measure of the separation between the projections is the difference between the means:

$$|w \cdot (m_a - m_b)|$$

and

$$m_a = \frac{1}{N} \sum_{i=1}^N x_i, \text{ summed over the } N \text{ } x \text{'s from class } a$$

$$m_b = \frac{1}{M} \sum_{j=1}^M x_j, \text{ summed over the } M \text{ } x \text{'s from class } b$$

where  $w$  (**wnvec**) is the unknown unit vector along the discriminant line .

In our case above, the vector difference between the means is:

```
Mean[a] - Mean[b]
```

```
{0.0831499, -1.92741}
```

and the difference between the means projected onto a discriminant line is:

$$\mathbf{wnvec} \cdot (\mathbf{Mean}[\mathbf{a}] - \mathbf{Mean}[\mathbf{b}])$$

1.13832

To improve separation, we can't just scale  $w$ , because the noise scales too.

We'd like the difference between the means to be large relative to the variation for each class. We can define a measure of the scatter for the projected samples in say class  $a$  ( $a=1$ ), by:

$$\sum_{y \in \text{class } a} (y - \tilde{m}_a)^2$$

where  $\tilde{m}_a$  is the sample mean of the points from class  $a$  projected onto discriminant line and  $y = \mathbf{w} \cdot \mathbf{x}$

Or in terms of the Mathematic example:

**Apply[Plus, proj - wnvec.Mean[a]];**

The total scatter  $S$  is defined by the sum of the scatters for both classes ( $a$  and  $b$ ).

$$S = \sum_{y \in \text{class } a} (y - \tilde{m}_a)^2 + \sum_{y \in \text{class } b} (y - \tilde{m}_b)^2$$

If we divide the above number by the total number of points, we have an estimate of the variance of the combined data along the projected axis.

We now have the basic ingredients behind intuition for the Fisher linear discriminant. We'd like to find that  $\mathbf{w}$  for which  $J$ :

$$J(\mathbf{w}) = \frac{|\tilde{m}_a - \tilde{m}_b|^2}{S} = \frac{|\mathbf{w} \cdot (\mathbf{m}_a - \mathbf{m}_b)|^2}{S}$$

is biggest. We want to maximize the difference between the projected class means, while minimizing the dispersion of the data on the projected line.

One can show that  $S = \mathbf{w}^T \cdot \mathbf{S}_W \cdot \mathbf{w}$ , where

$\mathbf{S}_W$  is measure of within-class variation called the *within-class scatter matrix*:

$$\mathbf{S}_W = \sum_{i=1}^2 \sum_{x \in \text{Class } i} (x - m_i)(x - m_i)^T$$

For the numerator, a measure of between class variation is the *between-class scatter matrix*:

$$\mathbf{S}_B = (\mathbf{m}_1 - \mathbf{m}_2) \cdot (\mathbf{m}_1 - \mathbf{m}_2)^T$$

and the difference between the projected means can be show to be:

$$|\tilde{m}_a - \tilde{m}_b|^2 = \mathbf{w}^T \cdot \mathbf{S}_B \cdot \mathbf{w}$$

Find  $\mathbf{w}$  (corresponding to slope) to maximize the criterion function

$$J(\mathbf{w}) = \frac{\mathbf{w}^T \cdot \mathbf{S}_B \cdot \mathbf{w}}{\mathbf{w}^T \cdot \mathbf{S}_W \cdot \mathbf{w}}$$

Answer:

$$\mathbf{w} = \mathbf{S}_W^{-1} \cdot (\mathbf{m}_a - \mathbf{m}_b)$$



## Demo: Finding Fisher's linear discriminant

```


normalize[x_] := x / Sqrt[x.x];

ma = Mean[a];
mb = Mean[b];

Sa = Sum[Outer[Times, a[[i]] - ma, a[[i]] - ma], {i, 1, nsamples}];
Sb = Sum[Outer[Times, b[[i]] - mb, b[[i]] - mb], {i, 1, nsamples}];
Sw = Sa + Sb;
wldf = normalize[Inverse[Sw].(ma - mb)]
{0.697928, -0.716168}

aproj = (#1 wldf &) /@ (a.wldf);
aproj = ListPlot[aproj, AspectRatio -> 1,
  PlotStyle -> Hue[0.3`], DisplayFunction -> Identity];
bproj = (#1 wldf &) /@ (b.wldf);
gbproj = ListPlot[bproj, AspectRatio -> 1,
  PlotStyle -> Hue[0.7`], DisplayFunction -> Identity];
Show[ga, gb, gaproj, gbproj, DisplayFunction -> $DisplayFunction]


```



We started off with a 2-dimensional input problem and turned it into a 1-D problem. For the n-dimensional case, see Duda and Hart.

## Compare with the principal component axes

```
c = Join[a, b];  
ListPlot[c, PlotRange -> {{-8, 8}, {-8, 8}}
```




```
g1 = ListPlot[c, PlotRange -> {{-8, 8}, {-8, 8}},  
AspectRatio -> 1, DisplayFunction -> Identity];
```

```
auto = Covariance[c]  
eigvalues = Eigenvalues[auto]  
eigauto = Eigenvectors[auto]  
{1.09252, 0.999432}, {0.999432, 2.48674}  
  
{3.00816, 0.571092}  
  
{0.462554, 0.886591}, {-0.886591, 0.462554}
```

```
gPCA = Plot[{eigauto[[1,2]]/eigauto[[1,1]] x,  
eigauto[[2,2]]/eigauto[[2,1]] x},  
{x,-4,4}, AspectRatio->1,  
DisplayFunction->Identity,  
PlotStyle->{RGBColor[.2,0,1]};
```

```
Show[g1, gPCA, DisplayFunction -> $DisplayFunction]
```



How does the principal component (biggest variance) compare with the Fisher discriminant line?

## References

- Duda, R. O., & Hart, P. E. (1973). *Pattern classification and scene analysis*. New York.: John Wiley & Sons.
- Duda, R. O., Hart, P. E., & Stork, D. G. (2001). *Pattern classification* (2nd ed.). New York: Wiley. (Amazon.com)
- Geman, S., Bienenstock, E., & Doursat, R. (1992). Neural networks and the bias/variance dilemma. *Neural Computation*, 4(1), 1-58.
- Mackay, D. J. C. (1992). Bayesian interpolation. *Neural Computation*, 4(3), 415-447.
- Ripley, B. D. (1996). *Pattern Recognition and Neural Networks*. Cambridge, UK: Cambridge University Press.
- Vapnik, V. N. (1995). *The nature of statistical learning*. New York: Springer-Verlag.  
<http://neuron.eng.wayne.edu/software.html>